// 编译：240884432
// 简称: vip11
// 名称: vip11
// 类别: 公式应用
// 类型: 用户应用
// 输出: 公众号松鼠Quant 主页：quant789.com
//------------------------------------------------------------------------
Params
	Numeric Length(20);//小区间参数5-100
	Numeric M(20);//均线参数 5-200
	Numeric smoothingFactor(0.5); //STC平滑系数，0.1-0.9
	Numeric TRS(75);//移动止损止盈幅度 参数范围及步长： 5-200，5
	Numeric Fund(20000); //投入保证金；
Vars
	Numeric Lots(0);
	//过滤变量
	Series<Numeric> MA1;
	Series<Numeric> MA2;
	//MACD默认常数
	Numeric FastLength(12);
	Numeric SlowLength(26);
	Numeric MACDLength(9);

	Numeric MACDDiff; 
	Numeric AvgMACD;
	
	//考夫指标变量
	Series<Numeric> MACDValue;
	Series<Numeric> lowestMACD;
	Series<Numeric> highestMACD;
	Series<Numeric> normalizedMACD;
	Series<Numeric> smoothedMACD(0);
	Series<Numeric> lowestSmoothedMACD;
	Series<Numeric> highestSmoothedMACD;
	Series<Numeric> smoothedNormalizedMACD;
	Series<Numeric> STCValue(0);
	Series<Numeric> STCValue_A(0);
	Series<Bool> cond1;
	Series<Bool> cond2;
	Series<Bool> cond3;
	Series<Bool> cond4;
	
	//出场变量
	Series<Numeric> MyPrice;//开仓价格
	Series<Numeric> MyPrice2;//开仓价格
	Series<Numeric> HighAfterEntry;//开仓后出现的最高价
	Series<Numeric> LowAfterEntry;//开仓后出现的最低价
	Series<Numeric> liQKA;
		//Bar更新事件函数，参数indexs表示变化的数据源图层ID248844320数组
	Series<Numeric> DliqPoint;
	Series<Numeric> KliqPoint;
	Series<Numeric> barcoutN;
	Series<Numeric> bar_entry_count;
	Series<Numeric> kaicang_kg;
	Series<Numeric> gobro(0);
	Series<Numeric> HH;
	Series<Numeric> LL;
Events
	//此处实现事件函数
	
	//初始化事件函数，策略运行期间，首先运行且只有一次，应用在订阅数据等操作
	OnInit()
	{
		//与数据源有关
		Range[0:DataCount-1]
		{
			//=========数据源相关设置==============
			AddDataFlag(Enum_Data_RolloverBackWard());	//设置后复权

			AddDataFlag(Enum_Data_RolloverRealPrice());	//设置映射真实价格

			AddDataFlag(Enum_Data_AutoSwapPosition());	//设置自动换仓

			//AddDataFlag(Enum_Data_IgnoreSwapSignalCalc());	//设置忽略换仓信号计算
		}

	}

	//Bar更新事件函数，参数indexs表示变化的数据源图层ID数组
	OnBar(ArrayRef<Integer> indexs)
	{		
			//计算手数，FUND是投入的保证金，最低下单一手
			Lots = Max(1, Round(Fund/(O*ContractUnit*BigPointValue* MarginRatio/rollover), 0));
			MA1=XAverage(C,M);
			MA2=XAverage(C,M*2);
			/*MACD与考夫指标代码计算代码-----------------------------------------------------------------------------------------------------------------------------------*/
		    Numeric a1 = 2 / (fastLength + 1);
			Numeric a2 = 2 / (slowLength + 1);
			//计算相关系数
			Numeric r2 = 0.5 * Power(correlation(close, BarCount, MACDLength), 2) + 0.5;
			Numeric K = r2 * ((1 - a1) * (1 - a2)) + (1 - r2) * ((1 - a1) / (1 - a2));
			//计算MACD值
			MACDValue = (close - close[1]) * (a1 - a2) + (-a2 - a1 + 2) * MACDValue[1] - K * MACDValue[2];		
				
			//计算MACD的最低值和最高值，计算他们的波动差
			lowestMACD = lowest(MACDValue, fastLength);
			highestMACD = highest(MACDValue, fastLength)- lowestMACD; //
			//归一化MACD
			normalizedMACD = IIF( highestMACD > 0,(MACDValue - lowestMACD) / highestMACD * 100,normalizedMACD);
			//平滑MACD
			smoothedMACD = IIF(smoothedMACD==0,normalizedMACD,smoothedMACD[1] + smoothingFactor * (normalizedMACD - smoothedMACD[1]));
			//计算平滑后MACD的最低值和最高值，计算他们的波动差
			lowestSmoothedMACD = lowest(smoothedMACD, fastLength);
			highestSmoothedMACD = highest(smoothedMACD, fastLength)- lowestSmoothedMACD; // 
			//归一化平滑后MACD
			smoothedNormalizedMACD = IIF(highestSmoothedMACD > 0,(smoothedMACD - lowestSmoothedMACD) / highestSmoothedMACD * 100 ,smoothedNormalizedMACD);
			//平滑归一化后MACD，得到STC值
			STCValue = IIF(STCValue==0 ,smoothedNormalizedMACD,STCValue[1] + smoothingFactor * (smoothedNormalizedMACD - STCValue[1]));
			STCValue_A=round(STCValue-50,0);
			//平滑MACD值
			Numeric MACDValuea= (MACDValue) / (XAverage(high - low, slowLength)) * 100;
			MACDValuea= ((MACDValuea) - XAverage(MACDValuea, 9))/2;
			
			/*画线代码-----------------------------------------------------------------------------------------------------------------------------------*/
			PlotNumeric("STCValue",STCValue_A);
			If (MACDValuea >= 0)	 
				PlotAuto("MACD",MACDValuea,0,Red,Enum_Bar); 
			Else
				PlotAuto("MACD",MACDValuea,0,Green,Enum_Bar); 
			PlotNumeric("0",0,0,red);
			
			/*开平仓条件代码-----------------------------------------------------------------------------------------------------------------------------------*/
			cond1=STCValue_A>STCValue_A[1] and STCValue_A[2]>=STCValue_A[1];
			cond2=STCValue_A<STCValue_A[1] and STCValue_A[2]<=STCValue_A[1];
			cond3=MA1>MA2 and C>MA2;
			cond4=MA1<MA2 and C<MA2;
			
			if(STCValue_A<0 and cond1)
			{
				PlotNumeric("buy",STCValue_A,STCValue_A,red);
			}
			
			if(STCValue_A>0 and cond2)
			{
				PlotNumeric("short",STCValue_A,STCValue_A,Green);
			}
			
			
			if(cond1)
			{
				gobro=1;
			}
			if(cond2)
			{
				gobro=-1;
			}
			HH=Highest(H,Length);
			LL=Lowest(L,Length);
			if(gobro[1]==1 and H>=HH[1] and STCValue_A[1]<0 and cond3[1]  and MarketPosition!=1)
			{
				Buy(Lots,max(open,HH[1]));
				LowAfterEntry = max(open,HH[1]);//保存多头开仓价格;
				kaicang_kg=1;   // 标记已开仓
				bar_entry_count=0;  // 重置连续未开仓的Bar计数
				barcoutN=0; //重置liQKA系数的计算
				liQKA = 1; 
			}
			if(gobro[1]==-1 and L<=LL[1] and STCValue_A[1]>0 and cond4[1]  and MarketPosition!=-1)
			{
				SellShort(Lots,min(open,LL[1]));
				HighAfterEntry = min(open,LL[1]);//保存空头开仓价格;
				kaicang_kg=1;   // 标记已开仓
				bar_entry_count=0;  // 重置连续未开仓的Bar计数
				barcoutN=0; //重置liQKA系数的计算
				liQKA = 1; 
			}
			
			/*出场模块--------------------------------------------------------------------------------------------------------------------------------------*/
			if(kaicang_kg[1] ==1) // 如果在本周期内有开仓操作
			{
				bar_entry_count=bar_entry_count+1; // 将开仓计数器加一
			}
			Else
			{
				bar_entry_count=0; // 如果没有开仓，则将开仓计数器重置为 0
			}
			
			//移动出场
			If(MarketPosition == 0)   // 自适应参数默认值；
			{
				liQKA = 1; 
				barcoutN=0;
			}Else if(bar_entry_count>barcoutN)					 //当有持仓的情况下，liQKA会随着持仓时间的增加而逐渐减小，即止损止盈幅度乘数的减少。
			{
				liQKA = liQKA - 0.1; 
				liQKA = Max(liQKA,0.3);
				barcoutN=bar_entry_count;
			}
			
			Commentary("kaicang_kg[1]"+text(kaicang_kg[1]));
			Commentary("bar_entry_count"+text(bar_entry_count));
			Commentary("barcoutN"+text(barcoutN));
			Commentary("liQKA"+text(liQKA));
			// 记录开仓后的最高点和最低点
			If(bar_entry_count > 0)
			{
				HighAfterEntry = Min(HighAfterEntry,High); // 空头止损，更新最低的最高价
				LowAfterEntry = Max(LowAfterEntry,Low);    // 多头止损，更新最高的最低价
			}
			
			if(MarketPosition>0)
			{
				DliqPoint = LowAfterEntry - (Open*TRS/1000)*liQKA; //经过计算，这根吊灯出场线会随着持仓时间的增加变的越来越敏感；
			}
			if(MarketPosition<0)
			{
				KliqPoint = HighAfterEntry + (Open*TRS/1000)*liQKA; //经过计算，这根吊灯出场线会随着持仓时间的增加变的越来越敏感；
			}
		
			//If(KliqPoint[1]>0 and MarketPosition<0)PlotNumeric("KliqPoint[1]",KliqPoint[1]);
			//if(DliqPoint[1]>0 and MarketPosition>0)PlotNumeric("DliqPoint[1]",DliqPoint[1]);
			
			// 持有多单时
			If(MarketPosition >0 And bar_entry_count >1  And Low <= DliqPoint[1] and DliqPoint[1]>0 and bar_entry_count>0) 
			{
			
					Sell(0,Min(Open,DliqPoint[1]));
					DliqPoint=0;
					kaicang_kg=0;
					gobro=0;
			}
			
			// 持有空单时
			If(MarketPosition <0 And bar_entry_count >1  And High >= KliqPoint[1] and KliqPoint[1]>0 and bar_entry_count>0)
			{	
				
					BuyToCover(0,Max(Open,KliqPoint[1]));
					KliqPoint=0;
					kaicang_kg=0;
					gobro=0;
			}
			
			
	}

